Rules for integrands of the form $P[x]\ (f\,x)^m\ (d+e\,x^2)^q\ (a+b\,x^2+c\,x^4)^p$

1. $\int \frac{x^m\,(A+B\,x^2+C\,x^4)}{(d+e\,x^2)\,\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ \frac{m}{2}\in\mathbb{Z}$

1: $\int \frac{x^m\,(A+B\,x^2+C\,x^4)}{(d+e\,x^2)\,\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ \frac{m}{2}\in\mathbb{Z}^+$

Rule: If $b^2-4\,a\,c\neq 0\ \wedge\ \frac{m}{2}\in\mathbb{Z}^+$, then

$$\int \frac{x^m\,(A+B\,x^2+C\,x^4)}{(d+e\,x^2)\,\sqrt{a+b\,x^2+c\,x^4}}\,dx\ \longrightarrow$$

$$\frac{C\,x^{m-1}\,\sqrt{a+b\,x^2+c\,x^4}}{c\,e\,(m+1)}-\frac{1}{c\,e\,(m+1)}\int\frac{x^{m-2}}{(d+e\,x^2)\,\sqrt{a+b\,x^2+c\,x^4}}\cdot$$
$$\left(a\,C\,d\,(m-1)-(A\,c\,e\,(m+1)-C\,(a\,e\,(m-1)+b\,d\,m))\,x^2-(B\,c\,e\,(m+1)-C\,(b\,e\,m+c\,d\,(m+1)))\,x^4\right)\,dx$$

Program code:

```
Int[Px_*x_^m_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,2],C=Coeff[Px,x,4]},
  C*x^(m-1)*Sqrt[a+b*x^2+c*x^4]/(c*e*(m+1)) -
  1/(c*e*(m+1))*Int[(x^(m-2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]))*
    Simp[a*C*d*(m-1)-(A*c*e*(m+1)-C*(a*e*(m-1)+b*d*m))*x^2-(B*c*e*(m+1)-C*(b*e*m+c*d*(m+1)))*x^4,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x^2,2] && NeQ[b^2-4*a*c,0] && IGtQ[m/2,0]
```

```
Int[Px_*x_^m_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,2],C=Coeff[Px,x,4]},
  C*x^(m-1)*Sqrt[a+c*x^4]/(c*e*(m+1)) -
  1/(c*e*(m+1))*Int[(x^(m-2)/((d+e*x^2)*Sqrt[a+c*x^4]))*
    Simp[a*C*d*(m-1)-(A*c*e*(m+1)-C*a*e*(m-1))*x^2-(B*c*e*(m+1)-C*c*d*(m+1))*x^4,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x^2,2] && IGtQ[m/2,0]
```

2: $\int \frac{x^m \left(A + B x^2 + C x^4\right)}{\left(d + e x^2\right) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \wedge \frac{m}{2} \in \mathbb{Z}^-$

Rule: If $b^2 - 4 a c \neq 0 \wedge \frac{m}{2} \in \mathbb{Z}^-$, then

$$\int \frac{x^m \left(A + B x^2 + C x^4\right)}{\left(d + e x^2\right) \sqrt{a + b x^2 + c x^4}} \, dx \longrightarrow$$

$$\frac{A x^{m+1} \sqrt{a + b x^2 + c x^4}}{a d (m + 1)} + \frac{1}{a d (m + 1)} \int \frac{x^{m+2}}{\left(d + e x^2\right) \sqrt{a + b x^2 + c x^4}} \cdot \left(a B d (m + 1) - A (a e (m + 1) + b d (m + 2)) + (a C d (m + 1) - A (b e (m + 2) + c d (m + 3))) x^2 - A c e (m + 3) x^4\right) dx$$

Program code:

```
Int[Px_*x_^m_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,2],C=Coeff[Px,x,4]},
  A*x^(m+1)*Sqrt[a+b*x^2+c*x^4]/(a*d*(m+1)) +
  1/(a*d*(m+1))*Int[(x^(m+2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]))*
     Simp[a*B*d*(m+1)-A*(a*e*(m+1)+b*d*(m+2))+(a*C*d*(m+1)-A*(b*e*(m+2)+c*d*(m+3)))*x^2-A*c*e*(m+3)*x^4,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x^2,2] && NeQ[b^2-4*a*c,0] && ILtQ[m/2,0]
```

```
Int[Px_*x_^m_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[Px,x,0],B=Coeff[Px,x,2],C=Coeff[Px,x,4]},
  A*x^(m+1)*Sqrt[a+c*x^4]/(a*d*(m+1)) +
  1/(a*d*(m+1))*Int[(x^(m+2)/((d+e*x^2)*Sqrt[a+c*x^4]))*
     Simp[a*B*d*(m+1)-A*a*e*(m+1)+(a*C*d*(m+1)-A*c*d*(m+3))*x^2-A*c*e*(m+3)*x^4,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x^2,2] && ILtQ[m/2,0]
```

Rules for integrands of the form $P[x]\ (d+e\,x^2)^q\ (a+b\,x^2+c\,x^4)^p$

1: $\int x\,P[x^2]\ (d+e\,x^2)^q\ (a+b\,x^2+c\,x^4)^p\,dx$

Derivation: Integration by substitution

Basis: $x\,F[x^2] = \frac{1}{2}\,\text{Subst}[F[x],\,x,\,x^2]\ \partial_x x^2$

Rule 1.2.2.7.1:

$$\int x\,P[x^2]\ (d+e\,x^2)^q\ (a+b\,x^2+c\,x^4)^p\,dx \longrightarrow \frac{1}{2}\,\text{Subst}\Big[\int P[x]\ (d+e\,x)^q\ (a+b\,x+c\,x^2)^p\,dx,\ x,\ x^2\Big]$$

Program code:

```
Int[x_*Px_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/2*Subst[Int[ReplaceAll[Px,x→Sqrt[x]]*(d+e*x)^q*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,d,e,p,q},x] && PolyQ[Px,x^2]
```

2: $\int P_r[x] \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p \, dx$ when PolynomialRemainder$[P_r[x], x, x] == 0$

– Derivation: Algebraic simplification

– Rule 1.2.2.7.2: If PolynomialRemainder$[P_r[x], x, x] == 0$, then

$$\int P_r[x] \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p \, dx \rightarrow \int x \, \text{PolynomialQuotient}[P_r[x], x, x] \left(d+e x^2\right)^q \left(a+b x^2+c x^4\right)^p \, dx$$

– Program code:

```
Int[Pr_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pr,x,x]*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x] && PolyQ[Pr,x] && EqQ[PolynomialRemainder[Pr,x,x],0] && Not[MatchQ[Pr,x^m_.*u_. /; IntegerQ[m]]]
```

3: $\int P_r[x]\,(d+e x^2)^q\,(a+b x^2+c x^4)^p\,dx$ when $\neg P_r[x^2]$

Derivation: Algebraic expansion

Basis: $P_r[x] == \sum_{k=0}^{r/2} P_r[x, 2k]\, x^{2k} + x \sum_{k=0}^{(r-1)/2} P_r[x, 2k+1]\, x^{2k}$

Note: This rule transforms $P_r[x]$ into a sum of the form $Q_s[x^2] + x\,R_t[x^2]$.

Rule 1.2.2.7.3: If $\neg P_r[x^2]$, then

$$\int P_r[x]\,(d+e x^2)^q\,(a+b x^2+c x^4)^p\,dx \longrightarrow$$

$$\int \left(\sum_{k=0}^{\frac{r}{2}} P_r[x, 2k]\, x^{2k}\right) (d+e x^2)^q\,(a+b x^2+c x^4)^p\,dx + \int x \left(\sum_{k=0}^{\frac{r-1}{2}} P_r[x, 2k+1]\, x^{2k}\right) (d+e x^2)^q\,(a+b x^2+c x^4)^p\,dx$$

Program code:

```
Int[Pr_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Module[{r=Expon[Pr,x],k},
  Int[Sum[Coeff[Pr,x,2*k]*x^(2*k),{k,0,r/2}]*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x] +
  Int[x*Sum[Coeff[Pr,x,2*k+1]*x^(2*k),{k,0,(r-1)/2}]*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x]] /;
FreeQ[{a,b,c,d,e,p,q},x] && PolyQ[Pr,x] && Not[PolyQ[Pr,x^2]]
```

4. $\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0$

1: $\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $a+b\,z+c\,z^2==(d+e\,z)\left(\frac{a}{d}+\frac{c\,z}{e}\right)$

Rule 1.2.2.7.4.1: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\in\mathbb{Z}$, then

$$\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx\,\longrightarrow\,\int P[x^2]\,(d+e\,x^2)^{p+q}\left(\frac{a}{d}+\frac{c\,x^2}{e}\right)^p\,dx$$

Program code:

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[Px*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p] &&
  (PolyQ[Px,x^2] || MatchQ[Px,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[Px*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,q},x] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p] &&
  (PolyQ[Px,x^2] || MatchQ[Px,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

2: $\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2==0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c\,d^2-b\,d\,e+a\,e^2==0$, then $\partial_x\frac{(a+b\,x^2+c\,x^4)^p}{(d+e\,x^2)^p\left(\frac{a}{d}+\frac{c\,x^2}{e}\right)^p}==0$

Basis: If $c d^2 - b d e + a e^2 == 0$, then $\frac{(a+b x^2+c x^4)^p}{(d+e x^2)^p \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p} == \frac{(a+b x^2+c x^4)^{\text{FracPart}[p]}}{(d+e x^2)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^{\text{FracPart}[p]}}$

- Rule 1.2.2.7.4.2: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 == 0 \land p \notin \mathbb{Z}$, then

$$\int P[x^2] (d + e x^2)^q (a + b x^2 + c x^4)^p \, dx \to \frac{(a + b x^2 + c x^4)^{\text{FracPart}[p]}}{(d + e x^2)^{\text{FracPart}[p]} \left(\frac{a}{d} + \frac{c x^2}{e}\right)^{\text{FracPart}[p]}} \int P[x^2] (d + e x^2)^{p+q} \left(\frac{a}{d} + \frac{c x^2}{e}\right)^p dx$$

- Program code:

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+c*x^2/e)^FracPart[p])*
    Int[Px*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]] &&
  (PolyQ[Px,x^2] || MatchQ[Px,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_,x_Symbol] :=
  (a+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+c*x^2/e)^FracPart[p])*
    Int[Px*(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,p,q},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] &&
  (PolyQ[Px,x^2] || MatchQ[Px,(f_+g_.*x^2)^r_./;FreeQ[{f,g,r},x]])
```

5: $\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ p\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.2.7.5: If $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ q\in\mathbb{Z}\ \land\ p\in\mathbb{Z}$, then

$$\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx\ \longrightarrow\ \int \text{ExpandIntegrand}\left[P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p,\,x\right]dx$$

Program code:

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,e,q},x] && PolyQ[Px,x^2] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Px*(d+e*x^2)^q*(a+c*x^4)^p,x],x] /;
FreeQ[{a,c,d,e,q},x] && PolyQ[Px,x^2] && NeQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

6. $\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ p+\frac{1}{2}\in\mathbb{Z}\ \land\ q\in\mathbb{Z}$

1. $\int \frac{P[x^2]\,(d+e\,x^2)^q}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ q\in\mathbb{Z}$

1: $\int \frac{(d+e\,x^2)^q\,(A+B\,x^2+C\,x^4)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ q\in\mathbb{Z}^+$

Rule 1.2.2.7.6.1.1: If $b^2-4\,a\,c\neq 0\ \land\ c\,d^2-b\,d\,e+a\,e^2\neq 0\ \land\ q\in\mathbb{Z}^+$, then

$$\int \frac{(d+e\,x^2)^q\,(A+B\,x^2+C\,x^4)}{\sqrt{a+b\,x^2+c\,x^4}}\,dx\ \longrightarrow$$

$$\frac{C\,x\,\left(d+e\,x^2\right)^q\sqrt{a+b\,x^2+c\,x^4}}{c\,(2\,q+3)}+$$

$$\frac{1}{c\,(2\,q+3)}\int\frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\left(d+e\,x^2\right)^{q-1}\left(A\,c\,d\,(2\,q+3)-a\,C\,d+(c\,(B\,d+A\,e)\,(2\,q+3)-C\,(2\,b\,d+a\,e+2\,a\,e\,q))\,x^2+(B\,c\,e\,(2\,q+3)-2\,C\,(b\,e-c\,d\,q+b\,e\,q))\,x^4\right)dx$$

–

Program code:

```
Int[(d_+e_.*x_^2)^q_*P4x_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  C*x*(d+e*x^2)^q*Sqrt[a+b*x^2+c*x^4]/(c*(2*q+3)) +
  1/(c*(2*q+3))*Int[(d+e*x^2)^(q-1)/Sqrt[a+b*x^2+c*x^4]*
     Simp[A*c*d*(2*q+3)-a*C*d+(c*(B*d+A*e)*(2*q+3)-C*(2*b*d+a*e+2*a*e*q))*x^2+(B*c*e*(2*q+3)-2*C*(b*e-c*d*q+b*e*q))*x^4,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[P4x,x^2] && EqQ[Expon[P4x,x],4] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[q,0]
```

```
Int[(d_+e_.*x_^2)^q_*P4x_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  C*x*(d+e*x^2)^q*Sqrt[a+c*x^4]/(c*(2*q+3)) +
  1/(c*(2*q+3))*Int[(d+e*x^2)^(q-1)/Sqrt[a+c*x^4]*
     Simp[A*c*d*(2*q+3)-a*C*d+(c*(B*d+A*e)*(2*q+3)-a*C*e*(2*q+1))*x^2+(B*c*e*(2*q+3)+2*c*C*d*q)*x^4,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[P4x,x^2] && EqQ[Expon[P4x,x],4] && NeQ[c*d^2+a*e^2,0] && IGtQ[q,0]
```

2: $\int \frac{(d+e x^2)^q (A+B x^2+C x^4)}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q+1 \in \mathbb{Z}^-$

–

Rule 1.2.2.7.6.1.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q+1 \in \mathbb{Z}^-$, then

$$\int \frac{(d+e x^2)^q (A+B x^2+C x^4)}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow$$

$$-\frac{(C d^2-B d e+A e^2) x (d+e x^2)^{q+1} \sqrt{a+b x^2+c x^4}}{2 d (q+1) (c d^2-b d e+a e^2)}+\frac{1}{2 d (q+1) (c d^2-b d e+a e^2)} \int \frac{(d+e x^2)^{q+1}}{\sqrt{a+b x^2+c x^4}} \cdot$$
$$\left(a d (C d-B e)+A \left(a e^2 (2 q+3)+2 d (c d-b e) (q+1)\right)-2 ((B d-A e) (b e (q+2)-c d (q+1))-C d (b d+a e (q+1))) x^2+c \left(C d^2-B d e+A e^2\right) (2 q+5) x^4\right) dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*P4x_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -(C*d^2-B*d*e+A*e^2)*x*(d+e*x^2)^(q+1)*Sqrt[a+b*x^2+c*x^4]/(2*d*(q+1)*(c*d^2-b*d*e+a*e^2)) +
  1/(2*d*(q+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x^2)^(q+1)/Sqrt[a+b*x^2+c*x^4]*
    Simp[a*d*(C*d-B*e)+A*(a*e^2*(2*q+3)+2*d*(c*d-b*e)*(q+1))-
      2*((B*d-A*e)*(b*e*(q+2)-c*d*(q+1))-C*d*(b*d+a*e*(q+1)))*x^2+
      c*(C*d^2-B*d*e+A*e^2)*(2*q+5)*x^4,x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[P4x,x^2] && LeQ[Expon[P4x,x],4] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[q,-1]
```

```
Int[(d_+e_.*x_^2)^q_*P4x_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -(C*d^2-B*d*e+A*e^2)*x*(d+e*x^2)^(q+1)*Sqrt[a+c*x^4]/(2*d*(q+1)*(c*d^2+a*e^2)) +
  1/(2*d*(q+1)*(c*d^2+a*e^2))*Int[(d+e*x^2)^(q+1)/Sqrt[a+c*x^4]*
    Simp[a*d*(C*d-B*e)+A*(a*e^2*(2*q+3)+2*c*d^2*(q+1))+2*d*(B*c*d-A*c*e+a*C*e)*(q+1)*x^2+c*(C*d^2-B*d*e+A*e^2)*(2*q+5)*x^4,x],x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[P4x,x^2] && LeQ[Expon[P4x,x],4] && NeQ[c*d^2+a*e^2,0] && ILtQ[q,-1]
```

3. $\int \frac{P[x^2]}{(d+e x^2) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1. $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0$

1. $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 == 0$

1: $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 == 0 \land B d + A e == 0$

Derivation: Integration by substitution

Basis: If $c d^2 - a e^2 == 0 \land B d + A e == 0$, then $\frac{A+B x^2}{(d+e x^2) \sqrt{a+b x^2+c x^4}} == A \, \text{Subst}\left[\frac{1}{d-(b d-2 a e) x^2}, x, \frac{x}{\sqrt{a+b x^2+c x^4}}\right] \partial_x \frac{x}{\sqrt{a+b x^2+c x^4}}$

Rule 1.2.2.7.6.1.3.1.1.1: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 == 0 \land B d + A e == 0$, then

$$\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \longrightarrow A \, \text{Subst}\left[\int \frac{1}{d - (b d - 2 a e) x^2} \, dx, x, \frac{x}{\sqrt{a + b x^2 + c x^4}}\right]$$

Program code:

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  A*Subst[Int[1/(d-(b*d-2*a*e)*x^2),x],x,x/Sqrt[a+b*x^2+c*x^4]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[c*d^2-a*e^2,0] && EqQ[B*d+A*e,0]
```

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  A*Subst[Int[1/(d+2*a*e*x^2),x],x,x/Sqrt[a+c*x^4]] /;
FreeQ[{a,c,d,e,A,B},x] && NeQ[c*d^2+a*e^2,0] && EqQ[c*d^2-a*e^2,0] && EqQ[B*d+A*e,0]
```

2: $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 == 0 \land B d + A e \neq 0$

Derivation: Algebraic expansion

Rule 1.2.2.7.6.1.3.1.1.2: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 == 0 \land B d + A e \neq 0$, then

$$\int \frac{A + B x^2}{\left(d + e x^2\right) \sqrt{a + b x^2 + c x^4}} \, dx \longrightarrow \frac{B d + A e}{2 d e} \int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx - \frac{B d - A e}{2 d e} \int \frac{d - e x^2}{\left(d + e x^2\right) \sqrt{a + b x^2 + c x^4}} \, dx$$

–

Program code:

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  (B*d+A*e)/(2*d*e)*Int[1/Sqrt[a+b*x^2+c*x^4],x] -
  (B*d-A*e)/(2*d*e)*Int[(d-e*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[c*d^2-a*e^2,0] && NeQ[B*d+A*e,0]
```

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  (B*d+A*e)/(2*d*e)*Int[1/Sqrt[a+c*x^4],x] -
  (B*d-A*e)/(2*d*e)*Int[(d-e*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e,A,B},x] && NeQ[c*d^2+a*e^2,0] && EqQ[c*d^2-a*e^2,0] && NeQ[B*d+A*e,0]
```

2. $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $\sqrt{b^2 - 4 a c} \in \mathbb{R} \lor c A^2 - b A B + a B^2 == 0$

1: $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c A^2 - b A B + a B^2 == 0$

Derivation: Piecewise constant extraction

Basis: If $c A^2 - b A B + a B^2 == 0$, then $\partial_x \frac{\sqrt{A+B x^2} \sqrt{\frac{a}{A} + \frac{c x^2}{B}}}{\sqrt{a+b x^2+c x^4}} == 0$

Rule 1.2.2.7.6.1.3.1.2.1: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c A^2 - b A B + a B^2 == 0$, then

$$\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \to \frac{\sqrt{A + B x^2} \sqrt{\frac{a}{A} + \frac{c x^2}{B}}}{\sqrt{a + b x^2 + c x^4}} \int \frac{\sqrt{A + B x^2}}{(d + e x^2) \sqrt{\frac{a}{A} + \frac{c x^2}{B}}} \, dx$$

Program code:

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  Sqrt[A+B*x^2]*Sqrt[a/A+c*x^2/B]/Sqrt[a+b*x^2+c*x^4]*Int[Sqrt[A+B*x^2]/((d+e*x^2)*Sqrt[a/A+c*x^2/B]),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[c*A^2-b*A*B+a*B^2,0]
```

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  Sqrt[A+B*x^2]*Sqrt[a/A+c*x^2/B]/Sqrt[a+c*x^4]*Int[Sqrt[A+B*x^2]/((d+e*x^2)*Sqrt[a/A+c*x^2/B]),x] /;
FreeQ[{a,c,d,e,A,B},x] && NeQ[c*d^2+a*e^2,0] && EqQ[c*A^2+a*B^2,0]
```

2: $\int \frac{A+B x^2}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land c d^2-b d e+a e^2\neq 0 \land c A^2-b A B+a B^2\neq 0 \land \sqrt{b^2-4 a c} \in \mathbb{R}$

Derivation: Algebraic expansion

Note: If $q\to \sqrt{b^2-4 a c}$ and $c d^2-b d e+a e^2\neq 0$, then $2 a e-d (b+q)\neq 0$.

Rule 1.2.2.7.6.1.3.1.2.2: If $b^2-4 a c>0 \land c d^2-b d e+a e^2\neq 0 \land c A^2-b A B+a B^2\neq 0$, let $q\to \sqrt{b^2-4 a c}$, if $q\in \mathbb{R}$, then

$$\int \frac{A+B x^2}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx \to \frac{2 a B-A (b+q)}{2 a e-d (b+q)} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx-\frac{B d-A e}{2 a e-d (b+q)} \int \frac{2 a+(b+q) x^2}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Sqrt[b^2-4*a*c]},
  (2*a*B-A*(b+q))/(2*a*e-d*(b+q))*Int[1/Sqrt[a+b*x^2+c*x^4],x] -
  (B*d-A*e)/(2*a*e-d*(b+q))*Int[(2*a+(b+q)*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
 RationalQ[q]] /;
FreeQ[{a,b,c,d,e,A,B},x] && GtQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*A^2-b*A*B+a*B^2,0]
```

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Sqrt[-a*c]},
  (a*B-A*q)/(a*e-d*q)*Int[1/Sqrt[a+c*x^4],x] -
  (B*d-A*e)/(a*e-d*q)*Int[(a+q*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
 RationalQ[q]] /;
FreeQ[{a,c,d,e,A,B},x] && GtQ[-a*c,0] && EqQ[c*d^2+a*e^2,0] && NeQ[c*A^2+a*B^2,0]
```

3. $\int \frac{A+B x^2}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c d^2-a e^2\neq 0$

1. $\int \frac{A+B x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c d^2-a e^2\neq 0 \land \frac{c}{a}>0$

x: $\int \frac{A+B x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c d^2-a e^2\neq 0 \land \frac{c}{a}>0 \land c A^2-a B^2==0$

Rule 1.2.2.7.6.1.3.1.3.1.x: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c d^2-a e^2\neq 0 \land \frac{c}{a}>0 \land c A^2-a B^2==0$, let $q\to\sqrt{\frac{B}{A}}$, then

$$\int \frac{A+B x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow$$

$$-\frac{(B d-A e)\,\text{ArcTan}\left[\frac{\sqrt{-b+\frac{c d}{e}+\frac{a e}{d}}\;x}{\sqrt{a+b x^2+c x^4}}\right]}{2 d e\sqrt{-b+\frac{c d}{e}+\frac{a e}{d}}}+\frac{B q\,(c d^2-a e^2)\,(A+B x^2)\sqrt{\frac{A^2(a+b x^2+c x^4)}{a(A+B x^2)^2}}}{4 c d e\,(B d-A e)\sqrt{a+b x^2+c x^4}}\,\text{EllipticPi}\left[-\frac{(B d-A e)^2}{4 d e A B},\,2\,\text{ArcTan}[q x],\,\frac{1}{2}-\frac{b A}{4 a B}\right]$$

Program code:

```
(* Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[B/A,2]},
  -(B*d-A*e)*ArcTan[Rt[-b+c*d/e+a*e/d,2]*x/Sqrt[a+b*x^2+c*x^4]]/(2*d*e*Rt[-b+c*d/e+a*e/d,2]) +
  B*q*(c*d^2-a*e^2)*(A+B*x^2)*Sqrt[A^2*(a+b*x^2+c*x^4)/(a*(A+B*x^2)^2)]/(4*c*d*e*(B*d-A*e)*Sqrt[a+b*x^2+c*x^4])*
    EllipticPi[-(B*d-A*e)^2/(4*d*e*A*B),2*ArcTan[q*x],1/2-b*A/(4*a*B)]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a] && EqQ[c*A^2-a*B^2,0] *)
```

```
(* Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[B/A,2]},
  -(B*d-A*e)*ArcTan[Rt[c*d/e+a*e/d,2]*x/Sqrt[a+c*x^4]]/(2*d*e*Rt[c*d/e+a*e/d,2]) +
  B*q*(c*d^2-a*e^2)*(A+B*x^2)*Sqrt[A^2*(a+c*x^4)/(a*(A+B*x^2)^2)]/(4*c*d*e*(B*d-A*e)*Sqrt[a+c*x^4])*
    EllipticPi[-(B*d-A*e)^2/(4*d*e*A*B),2*ArcTan[q*x],1/2]] /;
FreeQ[{a,c,d,e,A,B},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a] && EqQ[c*A^2-a*B^2,0] *)
```

1: $\int \frac{A+B x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c d^2-a e^2\neq 0 \land \frac{c}{a}>0 \land c A^2-a B^2==0$

– Rule 1.2.2.7.6.1.3.1.3.1.1: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land c d^2-a e^2\neq 0 \land \frac{c}{a}>0 \land c A^2-a B^2==0$, let $q\to\sqrt{\frac{B}{A}}$, then

$$\int \frac{A+B x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow$$

$$-\frac{(B d-A e)\,\text{ArcTan}\left[\frac{\sqrt{-b+\frac{c d}{e}+\frac{a e}{d}}\,x}{\sqrt{a+b x^2+c x^4}}\right]}{2 d e\sqrt{-b+\frac{c d}{e}+\frac{a e}{d}}}+\frac{(B d+A e)\,(A+B x^2)\sqrt{\frac{A^2(a+b x^2+c x^4)}{a (A+B x^2)^2}}}{4 d e A q\sqrt{a+b x^2+c x^4}}\,\text{EllipticPi}\left[-\frac{(B d-A e)^2}{4 d e A B},\,2\,\text{ArcTan}[q x],\,\frac{1}{2}-\frac{b A}{4 a B}\right]$$

– Program code:

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[B/A,2]},
  -(B*d-A*e)*ArcTan[Rt[-b+c*d/e+a*e/d,2]*x/Sqrt[a+b*x^2+c*x^4]]/(2*d*e*Rt[-b+c*d/e+a*e/d,2]) +
  (B*d+A*e)*(A+B*x^2)*Sqrt[A^2*(a+b*x^2+c*x^4)/(a*(A+B*x^2)^2)]/(4*d*e*A*q*Sqrt[a+b*x^2+c*x^4])*
    EllipticPi[Cancel[-(B*d-A*e)^2/(4*d*e*A*B)],2*ArcTan[q*x],1/2-b*A/(4*a*B)]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a] && EqQ[c*A^2-a*B^2,0]
```

```
Int[(A_+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[B/A,2]},
  -(B*d-A*e)*ArcTan[Rt[c*d/e+a*e/d,2]*x/Sqrt[a+c*x^4]]/(2*d*e*Rt[c*d/e+a*e/d,2]) +
  (B*d+A*e)*(A+B*x^2)*Sqrt[A^2*(a+c*x^4)/(a*(A+B*x^2)^2)]/(4*d*e*A*q*Sqrt[a+c*x^4])*
    EllipticPi[Cancel[-(B*d-A*e)^2/(4*d*e*A*B)],2*ArcTan[q*x],1/2]] /;
FreeQ[{a,c,d,e,A,B},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a] && EqQ[c*A^2-a*B^2,0]
```

2: $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 \neq 0 \land \frac{c}{a} > 0 \land c A^2 - a B^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B x^2}{d+e x^2} == \frac{B-A q}{e-d q} - \frac{(B d-A e) (1+q x^2)}{(e-d q) (d+e x^2)}$

Rule 1.2.2.7.6.1.3.1.3.1.2: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 \neq 0 \land \frac{c}{a} > 0 \land c A^2 - a B^2 \neq 0$, let $q \to \sqrt{\frac{c}{a}}$, then

$$\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \longrightarrow$$
$$\frac{A (c d + a e q) - a B (e + d q)}{c d^2 - a e^2} \int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx + \frac{a (B d - A e) (e + d q)}{c d^2 - a e^2} \int \frac{1 + q x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$$

Program code:

```
Int[(A_.+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (A*(c*d+a*e*q)-a*B*(e+d*q))/(c*d^2-a*e^2)*Int[1/Sqrt[a+b*x^2+c*x^4],x] +
  a*(B*d-A*e)*(e+d*q)/(c*d^2-a*e^2)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a] && NeQ[c*A^2-a*B^2,0]
```

```
Int[(A_.+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (A*(c*d+a*e*q)-a*B*(e+d*q))/(c*d^2-a*e^2)*Int[1/Sqrt[a+c*x^4],x] +
  a*(B*d-A*e)*(e+d*q)/(c*d^2-a*e^2)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e,A,B},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a] && NeQ[c*A^2-a*B^2,0]
```

2: $\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 \neq 0 \land \frac{c}{a} \ngtr 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B x^2}{d+e x^2} == \frac{B}{e} + \frac{e A-d B}{e (d+e x^2)}$

Rule 1.2.2.7.6.1.3.1.3.2: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 \neq 0 \land \frac{c}{a} \ngtr 0$, then

$$\int \frac{A + B x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \to \frac{B}{e} \int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx + \frac{e A - d B}{e} \int \frac{1}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$$

Program code:

```
Int[(A_.+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  B/e*Int[1/Sqrt[a+b*x^2+c*x^4],x] + (e*A-d*B)/e*Int[1/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && NegQ[c/a]
```

```
Int[(A_.+B_.*x_^2)/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  B/e*Int[1/Sqrt[a+c*x^4],x] + (e*A-d*B)/e*Int[1/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e,A,B},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && NegQ[c/a]
```

2. $\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0$

1: $\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 == 0$

Derivation: Algebraic expansion

Rule 1.2.2.7.6.1.3.2.1: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 == 0$, then

$$\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \longrightarrow -\frac{C}{e^2} \int \frac{d - e x^2}{\sqrt{a + b x^2 + c x^4}} \, dx + \frac{1}{e^2} \int \frac{C d^2 + A e^2 + B e^2 x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$$

Program code:

```
Int[P4x_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -C/e^2*Int[(d-e*x^2)/Sqrt[a+b*x^2+c*x^4],x] +
  1/e^2*Int[(C*d^2+A*e^2+B*e^2*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[P4x,x^2,2] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[c*d^2-a*e^2,0]
```

```
Int[P4x_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -C/e^2*Int[(d-e*x^2)/Sqrt[a+c*x^4],x] +
  1/e^2*Int[(C*d^2+A*e^2+B*e^2*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[P4x,x^2,2] && NeQ[c*d^2+a*e^2,0] && EqQ[c*d^2-a*e^2,0]
```

2. $\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 \neq 0$

1: $\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 \neq 0 \land \frac{c}{a} > 0 \land b^2 - 4 a c \not> 0$

– Derivation: Algebraic expansion

– Rule 1.2.2.7.6.1.3.2.2.1: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land c d^2 - a e^2 \neq 0 \land \frac{c}{a} > 0 \land b^2 - 4 a c \not> 0$, let $q \to \sqrt{\frac{c}{a}}$, then

$$\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \to -\frac{C}{e q} \int \frac{1 - q x^2}{\sqrt{a + b x^2 + c x^4}} \, dx + \frac{1}{c e} \int \frac{A c e + a C d q + (B c e - C (c d - a e q)) x^2}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$$

– Program code:

```
Int[P4x_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2],A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -C/(e*q)*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x] +
  1/(c*e)*Int[(A*c*e+a*C*d*q+(B*c*e-C*(c*d-a*e*q))*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[P4x,x^2,2] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a] && Not[GtQ[b^
```

```
Int[P4x_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2],A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -C/(e*q)*Int[(1-q*x^2)/Sqrt[a+c*x^4],x] +
  1/(c*e)*Int[(A*c*e+a*C*d*q+(B*c*e-C*(c*d-a*e*q))*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[P4x,x^2,2] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a]
```

2: $\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0 \wedge c d^2 - a e^2 \neq 0$

Derivation: Algebraic expansion (polynomial division)

Rule 1.2.2.7.6.1.3.2.2.2: If $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0 \wedge c d^2 - a e^2 \neq 0$, then

$$\int \frac{A + B x^2 + C x^4}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \to -\frac{1}{e^2} \int \frac{C d - B e - C e x^2}{\sqrt{a + b x^2 + c x^4}} \, dx + \frac{C d^2 - B d e + A e^2}{e^2} \int \frac{1}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$$

Program code:

```
Int[P4x_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -1/e^2*Int[(C*d-B*e-C*e*x^2)/Sqrt[a+b*x^2+c*x^4],x] +
  (C*d^2-B*d*e+A*e^2)/e^2*Int[1/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[P4x,x^2,2] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*d^2-a*e^2,0]
```

```
Int[P4x_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{A=Coeff[P4x,x,0],B=Coeff[P4x,x,2],C=Coeff[P4x,x,4]},
  -1/e^2*Int[(C*d-B*e-C*e*x^2)/Sqrt[a+c*x^4],x] +
  (C*d^2-B*d*e+A*e^2)/e^2*Int[1/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && PolyQ[P4x,x^2,2] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0]
```

3: $\int \frac{P_q[x]}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0 \wedge q > 4$

Rule 1.2.2.7.6.1.3.3: If $b^2 - 4 a c \neq 0 \wedge c d^2 - b d e + a e^2 \neq 0 \wedge q > 4$, then

$$\int \frac{P_q[x]}{(d + e x^2) \sqrt{a + b x^2 + c x^4}} \, dx \to$$

$$\frac{P_q[x,q]\,x^{q-5}\sqrt{a+b x^2+c x^4}}{c\,e\,(q-3)}+\frac{1}{c\,e\,(q-3)}\int\frac{c\,e\,(q-3)\,P_q[x]-P_q[x,q]\,x^{q-6}\left(d+e x^2\right)\left(a\,(q-5)+b\,(q-4)\,x^2+c\,(q-3)\,x^4\right)}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[Px_/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Expon[Px,x]},
  Coeff[Px,x,q]*x^(q-5)*Sqrt[a+b*x^2+c*x^4]/(c*e*(q-3)) +
  1/(c*e*(q-3))*
    Int[(c*e*(q-3)*Px-Coeff[Px,x,q]*x^(q-6)*(d+e*x^2)*(a*(q-5)+b*(q-4)*x^2+c*(q-3)*x^4))/
      ((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
 GtQ[q,4]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[Px_/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Expon[Px,x]},
  Coeff[Px,x,q]*x^(q-5)*Sqrt[a+c*x^4]/(c*e*(q-3)) +
  1/(c*e*(q-3))*
    Int[(c*e*(q-3)*Px-Coeff[Px,x,q]*x^(q-6)*(d+e*x^2)*(a*(q-5)+c*(q-3)*x^4))/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
 GtQ[q,4]] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x] && NeQ[c*d^2+a*e^2,0]
```

x: $\int\frac{P_q[x^2]\left(a+b x^2+c x^4\right)^p}{d+e x^2}\,dx$ when $b^2-4ac\neq 0\land c d^2-b d e+a e^2\neq 0\land p<-1$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.2.7.6.x: If $b^2-4ac\neq 0\land c d^2-b d e+a e^2\neq 0\land p<-1$, let $Q_{q-2}[x^2]\to \text{PolynomialQuotient}[P_q[x^2],\ a+b x^2+c x^4,\ x]$ and $A+B x^2\to\text{PolynomialRemainder}[P_q[x^2],\ a+b x^2+c x^4,\ x]$, then

$$\int\frac{P_q[x^2]\left(a+b x^2+c x^4\right)^p}{d+e x^2}\,dx\longrightarrow$$

$$\frac{B}{e}\int\left(a+b x^2+c x^4\right)^p dx-\frac{B d-A e}{e}\int\frac{\left(a+b x^2+c x^4\right)^p}{d+e x^2}dx+\int\frac{Q_{q-2}\left[x^2\right]\left(a+b x^2+c x^4\right)^{p+1}}{d+e x^2}dx\longrightarrow$$

$$-\frac{B x\left(b^2-2 a c+b c x^2\right)\left(a+b x^2+c x^4\right)^{p+1}}{2 a e (p+1)\left(b^2-4 a c\right)}+$$

$$\left((B d-A e) x\left(b^2 c d-2 a c^2 d-b^3 e+3 a b c e+c\left(b c d-b^2 e+2 a c e\right) x^2\right)\left(a+b x^2+c x^4\right)^{p+1}\right)/\left(2 a e (p+1)\left(b^2-4 a c\right)\left(c d^2-b d e+a e^2\right)\right)+$$

$$\int\frac{\left(a+b x^2+c x^4\right)^{p+1}}{d+e x^2}\left(\frac{P_q\left[x^2\right]}{a+b x^2+c x^4}-\frac{d+e x^2}{\left(a+b x^2+c x^4\right)^{p+1}}\cdot\right.$$

$$\partial_x\left(-\frac{B x\left(b^2-2 a c+b c x^2\right)\left(a+b x^2+c x^4\right)^{p+1}}{2 a e (p+1)\left(b^2-4 a c\right)}+\right.$$

$$\left.\left.\left((B d-A e) x\left(b^2 c d-2 a c^2 d-b^3 e+3 a b c e+c\left(b c d-b^2 e+2 a c e\right) x^2\right)\left(a+b x^2+c x^4\right)^{p+1}\right)/\left(2 a e (p+1)\left(b^2-4 a c\right)\left(c d^2-b d e+a e^2\right)\right)\right)\right)dx$$

Program code:

```
(* Int[Pq_*(a_+b_.*x_^2+c_.*x_^4)^p_/(d_+e_.*x_^2),x_Symbol] :=
  With[{A=Coeff[PolynomialRemainder[Pq,a+b*x^2+c*x^4,x],x,0],
        B=Coeff[PolynomialRemainder[Pq,a+b*x^2+c*x^4,x],x,2]},
  -B*x*(b^2-2*a*c+b*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*a*e*(p+1)*(b^2-4*a*c)) +
  (B*d-A*e)*x*(b^2*c*d-2*a*c^2*d-b^3*e+3*a*b*c*e+c*(b*c*d-b^2*e+2*a*c*e)*x^2)*(a+b*x^2+c*x^4)^(p+1)/
    (2*a*e*(p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)) +
  Int[(a+b*x^2+c*x^4)^(p+1)/(d+e*x^2)*ExpandToSum[Pq/(a+b*x^2+c*x^4)-(d+e*x^2)/(a+b*x^2+c*x^4)^(p+1)*
    D[-B*x*(b^2-2*a*c+b*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*a*e*(p+1)*(b^2-4*a*c)) +
      (B*d-A*e)*x*(b^2*c*d-2*a*c^2*d-b^3*e+3*a*b*c*e+c*(b*c*d-b^2*e+2*a*c*e)*x^2)*(a+b*x^2+c*x^4)^(p+1)/
        (2*a*e*(p+1)*(b^2-4*a*c)*(c*d^2-b*d*e+a*e^2)),x],x],x]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Pq,x^2] && Expon[Pq,x^2]>0 && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] *)
```

2: $\int P\left[x^2\right]\left(d+e x^2\right)^q\left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c\neq 0\wedge c d^2-b d e+a e^2\neq 0\wedge p+\frac{1}{2}\in\mathbb{Z}\wedge q\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.2.7.6.2: If $b^2-4 a c\neq 0\wedge c d^2-b d e+a e^2\neq 0\wedge p+\frac{1}{2}\in\mathbb{Z}\wedge q\in\mathbb{Z}$, then

$$\int P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx \longrightarrow \int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,\text{ExpandIntegrand}\left[P[x^2]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^{p+\frac{1}{2}},\,x\right]dx$$

Program code:

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[1/Sqrt[a+b*x^2+c*x^4],Px*(d+e*x^2)^q*(a+b*x^2+c*x^4)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[Px,x^2] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p+1/2] && IntegerQ[q]
```

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[1/Sqrt[a+c*x^4],Px*(d+e*x^2)^q*(a+c*x^4)^(p+1/2),x],x] /;
FreeQ[{a,c,d,e},x] && PolyQ[Px,x^2] && NeQ[c*d^2+a*e^2,0] && IntegerQ[p+1/2] && IntegerQ[q]
```

U: $\int P[x]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$

Rule 1.2.2.7.U:

$$\int P[x]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx \longrightarrow \int P[x]\,(d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$$

Program code:

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[Px*(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x] && PolyQ[Px,x]
```

```
Int[Px_*(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[Px*(d+e*x^2)^q*(a+c*x^4)^p,x] /;
FreeQ[{a,c,d,e,p,q},x] && PolyQ[Px,x]
```